
Week 10 - RATS Lecture – ARCH Models 
March 29, 2006 
 
Estimating an ARCH model involves using maximum likelihood estimation. 
 
Instead of doing: 
 
linreg cons / consr 
# constant pmsat 
  
Here is an example of a maximum likelihood regression with no ARCH model: 
 
nonlin b0 b1 var 
frml e = cons - b0 - b1*pmsat 
frml LA = -0.5*(log(var) + e(t)**2/var) 
compute b0 = 1.0, b1 = .5, var=3.0 
maximize(method=bfgs,trace,iterations=200) LA 1 154 
 
Which breaks down into: 
 
nonlin b0 b1 var 
 “nonlin” sets up a list of parameters in the model 
 we name 3 parameters to estimate here 
frml e = cons - b0 - b1*pmsat 
 “frml” is for a ml formula 
 “e” is for error. 
frml LA = -0.5*(log(var) + e(t)**2/var) 
 this sets up the likelihood function – OLS translated into log likelihood 
compute b0 = 1.0, b1 = .5, var=3.0 
 starting value guesses based on OLS.  Var = 3.0 comes from the squared 
 Standard Error of Estimate. 
maximize(method=bfgs,trace,iterations=200) LA 1 154 
 maximize the likelihood function LA for all 154 timepoints. 
 
We can put the linreg commands into the program to get exact starting values. 
 
nonlin b0 b1 var 
frml e = cons - b0 - b1*pmsat 
frml LA = -0.5*(log(var) + e(t)**2/var) 
linreg cons 
# constant pmsat   
com b0=%beta(1),b1=%beta(2),var=%seesq,a1=0.05 
maximize(method=bfgs,trace,iterations=200) LA 1 154 

Testing for ARCH 
 
We want to examine the squared residuals. 
 
set ressq = consr*consr 
 create new series of squared residuals 
correlate(number=20,qstats,dfc=2) ressq 
 look at the autocorrelation in the series of squared residuals 
@bjident ressq 
 look at the ACF and PACF of the series of squared residuals. 
 
Lagrange Multiplier Test: 
 
linreg ressq 
# constant ressq{1} 
 regress squared residuals on itself at various lags – here the ACF suggests just 
 the first lag. 
compute trsq = %nobs*%rsquared 
 compute the test statistic. 
cdf chisqr trsq 1 
 Is the test statistic significant – 1 is for 1 degree of freedom since p=1 
 
Now, estimate the ARCH model: 
 
nonlin c1 c2 a0 a1 
frml ex = cons - c1 -c2*pmsat 
frml h = a0 + a1*ex(t-1)**2 
frml L = -0.5*(log(h) + ex(t)**2/h) 
compute c1 = 15.0, c2 = 0.4, a0 = 8.0, a1 = 0.1 
maximize(method=bfgs,trace,iterations=200) L 2 154 
 
Line by line: 
 
nonlin c1 c2 a0 a1 
 give names to parameters to be estimated 
 c1 and c2 are parameters for the regression. a0 and a1 are parameters for ARCH 
frml ex = cons - c1 -c2*pmsat 
 equation for errors of main equation 
frml h = a0 + a1*ex(t-1)**2 
 this specifies the ARCH process.  a1 is the ARCH parameter.  We are multiplying 
 it by the squared error lagged back one period, that is, its an ARCH-1. 
frml L = -0.5*(log(h) + ex(t)**2/h) 
 sets up the likelihood function 
compute c1 = 15.0, c2 = 0.4, a0 = 1.0, a1 = 0.1 
 starting values  
maximize(method=bfgs,trace,iterations=200) L 2 154 

 maximize the likelihood function L.  for observations 2-154.  We begin with two 
 since we lost one by having a lag of the squared error. 
 default is 20 iterations. 
 “trace” shows us the iterations. 
 “method=bfgs” chooses the Broyden, Fletcher, Goldfarb and Shanno method 
 (described in Press, et al. 1988).  This is one of several MLE methods. 
 bhhh is another option that converges more quickly. 
 
Two other ways to build GARCH models are using the GARCHN.SRC procedure file or 
using the ARCH/GARCH Wizard in the menu of RATS. 
 
The source file is the older method. 
 
These are ok for simple univariate models but not great for complex models. 

